
CDF

CDF(BIVNORM or CHISQ or DICKEYF or F or NORMAL or T or WTDCHI,

DF=CHISQあるいはTに対する自由度, DF1=Fに対する分子の自由度,

DF2=Fに対する分母の自由度, NLAGS=拡張Dickey-Fuller検定のラグ数,

NOB=単位根検定あるいは共和分検定に対する観測値数,

NVAR=共和分検定に対する変数の数, RHO=BIVNORMに対する相関係数,

EIGVAL=WTDCHIに対する固有値ベクトル, LOWTAIL or UPTAIL or TWOTAIL,

CONSTANT, TREND, TSQ, INVERSE, PRINT)

検定統計量 [ 有意水準 ] ;

or

有意水準 [ 臨界値 ]; (INVERSEの時)

or

x の値 y の値 [ 有意水準 ]; (BIVNORMの時)

機能 :

CDFは裾の確率 (P -値あるいは有意水準), あるいはいくつかの累積分布関数を計算し,プリントします. こ

れは仮説検定に役立ちます.

使用法:

CDFに続いてスカラーの検定統計量の値を書くのが最も簡単なコマンドの形です. この場合, 正規分布に

対する両側確率が計算され,プリントされます. INVERSEオプションを用いる場合, 最初の引数は確率水

準でなければいけません. 臨界値が計算されます. その他の分布, あるいは裾の面積はオプションによって

選ぶことが出来ます. 回帰診断での仮説検定は, REGOPT(PVPRINT)コマンドを参照して下さい.

オプション:

BIVNORM/CHISQ/DICKEYF/F/NORMAL/T/WTDCHIは, それぞれ 2変量正規, カイ2乘 (�2), デ

ィッキー・フラー (DICKEY-FULLER), F , 標準正規, ステューデントの t, 加重カイ2乗分布を指示します.

DF=�2 あるいは t 分布の自由度, あるいはDickey-Fullerに対する観測値の数 (Dickey-Fullerに対する

NOB=オプションを参照).

DF1=F 分布の分子の自由度

DF2=F 分布の分母の自由度

RHO=2変量正規分布の相関係数

EIGVAL=加重 �2 分布の固有値ベクトル

NLAGS= 拡張Dickey=Fuller検定におけるラグ付階差の数. この数値は近似的な有限標本 P 値あるいは

臨界値を計算するのに用いられます. 既定値はゼロ (拡張しない検定を仮定)です. NOB=オプションは, 有

限標本値に対しても指定する必要があります.

NOB= 拡張Dickey=Fuller検定あるいはEngle-Granger検定に対する観測値の数. この数値は, 近似的な

有限標本 P 値あるいは臨界値を計算するのに用いられます. 既定値はゼロ (有限標本値の代わりに漸近的

な値を計算する).

NVAR=Engle-Granger/Dickey-Fuller共和分検定における変数の数. 既定値は 1(単なる単位根検定)で, 最

大6です.

CONSTANT/NOCONSTは定数項 (C)をDickey-Fullerの回帰に含めるかどうか指定します. NOCONST

だけがNVAR=1に対して有効です.

TREND/NOTREND はトレンド項 (1,2,. . .,T)をDickey-Fullerの回帰に含めるかどうか指定します.

TSQ/NOTSQは 2乗したトレンド項 (1,4,9,. . .)をDickey-Fullerの回帰に含めるかどうか指定します.

LOWTAIL/TWOTAIL/UPTAIL は密度関数の積分範囲を指定します. TWOTAILはほとんどの対

称分布 (正規と t)に対して既定値で, UPTAILは �2と F に対して, そしてLOWTAILは 2変量正規と

Dickey-Fullerに対して既定値です. TWOTAILは 2変量正規分布は定義できません.

INVERSE/NOINVERSE は逆分布関数 (インプットが有意水準, アウトプットが臨界値)を指定します.

通常, インプットは検定統計量でアウトプットが有意水準です. INVERSEは 2変量正規分布に対しては定

義されません.

PRINT/NOPRINTはスカラーのインプット変量をプリントします. PRINTはアウトプットを指定しな

ければ既定値になります.

アウトプット :

PRINTオプションが有効なら, インプット値とアウトプット値の両方が自由度と共にプリントされます. 2

番目の引数与えられていれば , アウトプット値が入れられ保存されます (例の 4と 6を参照). インプットと

アウトプットの引数はどんな数値のTSP変数でも構いません.

例:

1. 自由度が 5のハウスマン検定統計量の有意水準を計算する.

CDF(CHISQ,DF=5) HAUS;

この例のアウトプットは-- CHISQ(5) Test Statistic: 7.235999 , Upper tail area: .20367

2. ラグ付き従属変数のあるAR(1)に対する有意水準:

CDF @DHALT; or REGOPT(PVPRINT) DHALT; を回帰の前におく

3. 正規分布に対する両側臨界値

CDF(INV) .05;

この例のアウトプットは-- NORMAL Critical Value: 1.959964 , Two-tailed area: .05000

4. 正規分布のいくつかの臨界値

READ PX; .1 .05 .01;

CDF(INV,NORM) PX CRIT;

PRINT PX,CRIT;

5. F の臨界値

CDF(INV,F,DF1=3,DF2=10) .05;

この例のアウトプットは-- F(3,10) Critical Value: 3.708265 , Upper tailed area: .05000

6. 2変量正規

CDF(BIVNORM,RHO=.5,PRINT) -1 -2 PBIV;

この例のアウトプットは-- BIVNORM Test Statistic: -1.0000 , -2.0000 , Lower tail area: .01327

7. ディッキー・フラーの単位根検定 (UNIT(MAXLAG=0,NOWS) Y; と同じ)

TREND TIME;

SMPL 2,50;

DY = Y-Y(-1);

OLSQ DY TIME C Y(-1) ;

CDF(DICKEYF) @T(3);

8. 拡張ディッキー・フラーの単位根検定で, 有限標本 P-値 (UNIT(MAXLAG=3,NOWS,FINITE) Y;と

同じ)

TREND TIME;

SMPL 2,50;

DY = Y-Y(-1);

SMPL 5,50 ;

OLSQ DY TIME C Y(-1) DY(-1)-DY(-3);

CDF(DICKEYF,NOB=@NOB,NLAGS=3) @T(3);

9. エングル・グランジャーの共和分検定 (COINT(NOUNIT,MAXLAG=0,ALLORD) Y1-Y4;と同じ)

TREND TIME;

OLSQ Y1 C TIME Y2 Y3 Y4; EGTEST;

OLSQ Y2 C TIME Y1 Y3 Y4; EGTEST;

OLSQ Y3 C TIME Y1 Y2 Y4; EGTEST;

OLSQ Y4 C TIME Y1 Y2 Y3; EGTEST;

PROC EGTEST;

SMPL 2,50;

DU = @RES-@RES(-1);

OLSQ DU @RES(-1);

CDF(DICKEYF,NVAR=4) @T;

SMPL 1,50;

ENDPROC;

10. Durbin-Watson統計量の臨界値のチェック, 観測値10で 2つの右辺の変数:

SMPL 1,10; OLSQ Y C X1;

SET PI=4*ATAN(1); SET F=PI/(2*@NOB); TREND I;

EIGB=4*SIN(I*F)**2;

SELECT I<=(@NOB-@NCOEF); ?dLに対する最大固有値を用いる
CDF(WTDCHI,EIG=EIGB) .879; ?5%(n=10,k'=1) に対する dL

SELECT (@NCOEF<=I) & (I<=(@NOB-1)); ?dUに対する最小固有値を用いる
MMAKE dU 1.320 1.165 1.001; ?5%,2.5%,1%(n=10,k'=1) に対する dU

CDF(WTDCHI,EIG=EIGB,PRINT) dU;

11. Durbin-Watson統計量の厳密なP-値の再現 (REGOPTですでにできる)

SMPL 1,10;

READ Y X1; 4 2 7 4 7.5 6 4 3 2 1 3 2 5 3 4.5 4 7.5 8 5 6;

REGOPT(DWPVAL=EXACT);

OLSQ Y C X1;

MMAKE X @RNMS;

MAT XPXI=(X'X)";

TREND OBS; SELECT OBS>1;

DC=0; DX1=X1-X1(-1);

MMAKE DX DC DX1;

MMAKE BVEC 2 -1; MFORM(BAND,NROW=@NOB) DDP=BVEC;

MAT DMDP=DDP-DX*XPXI*DX';

? DMD'=D*D'-DX*(X'X)"(DX)' の固有値
? (A=D'D であるから MAの非ゼロ固有値と同じ)

MAT ED = EIGVAL(DMDP);

CDF(WTDCHI,EIG=ED) @DW;

方法:

BIVNORM: ACM Algorithm 462. Inverse は x あるいは y が既知でない限り一意でないので計算され

ない.

CHISQ: DCDFLIB 法：Abramovitz-Stegun 公式 26.4.19は不完全ガンマに変換し , 次にDiDinato and

Morris (1993)を用いる. Inverse はイタレーションによる (x の値を与えて p を得る{高速法も知られてい

る).

F と T : ACM Algorithm 322, あるいは DF1と DF2 の大きい場合 (DF1+DF2 > 500 でDF1 >1)には,

ACM Algorithm 346. Inverseはイタレーションによる.

NORMAL: ACM Algorithm 304, E < -37.5に対しては二次近似. Inverseは Applied Statistics Algorithm

AS241, from StatLib.

DICKEYF: 漸近値はMacKinnon(1994)の表3と 4による. 有限標本臨界値はCheung and Lai (1995) [Au-

gumented Dickey-Fuller]あるいは MacKnnon (1991)[Engle-Granger]から採用した. これらの数値を有限

標本臨界値へ変換するために, サイズ .05 と .01 あるいは .10 (どちらか観測された検定統計量に近い方)

でロジスティック補間を用いた. そのような P 値は .01, .05, あるいは .10のサイズで検定するためには十

分であるが, この範囲を超えるときわめて安定性がない.

WTDCHI: wt が (オプション EIGVALから得られた)固有値, ci は �2(1) 変数, そして d は (引数として得

られる)検定統計量とすると, WTDCHIは次の確率を計算します: Prob [
P
(wi � ci) < d] = Prob [

P
((wi�

d) � ci) < 0]. これはDurbin-Watson統計量や, 正規変量のその他の二次形式の比やいくつかの非入れ子型

検定 (たとえば , Vuong(1989)は非入れ子型仮説に対して尤度比検定を提案している)の P-値を計算するの

に便利です. Pan法は固有値の数が 90より小さい場合に用いられ, その他の場合は Imho�法が用いられま

す. もし最小固有値の絶対値が 1D-12より小さい場合は, それらを用いません. その他の二重の固有値は

チェックしません. この分布の逆関数は計算しません.

